
Certificat Informatique et Internet 
  

Feuille 7 
  

VBA : deuxième partie 

 

Le VBA (Visual Basic for Applications) est un langage proche du Visual Basic qui nécessite une application hôte pour 

s'exécuter (Excel dans notre cas). 

Grâce au VBA nous allons pouvoir réaliser à peu près tout ce que l'on souhaite avec Excel ... 

Mais avant de commencer, commençons par afficher les outils qui nous seront utiles. 

 

Site Web du cours : http://www.polytech.unice.fr/~hrenard/?page=enseignement#c2i 

1 Les conditions 

1.1 If 
Les conditions sont très utiles en programmation, elles nous serviront à effectuer des actions en fonction de critères 

précis (même principe que la fonction SI). 

La principale fonction est If, voici comment elle fonctionne : 

If [CONDITION ICI] Then ' => SI condition validée ALORS 

   'Instructions si vrai 

Else ' => SINON 

   'Instructions si faux 

End If 

 

Passons directement à la pratique et reprenons l'exemple développé à la leçon sur les variables. Il avait pour but 

d'afficher dans une boite de dialogue la ligne du tableau correspondant au numéro indiqué dans la cellule F5. : 

 
Si vous entrez une lettre en F5, cela génère un bug. Nous voulons éviter cela. 

http://www.polytech.unice.fr/~hrenard/?page=enseignement%23c2i

Sub variables() 

   'Déclaration des variables 

   Dim nom As String, prenom As String, age As Integer, numero_ligne As Integer 

         

   'Valeurs des variables 

   numero_ligne = Range("F5") + 1 

   nom = Cells(numero_ligne, 1) 

   prenom = Cells(numero_ligne, 2) 

   age = Cells(numero_ligne, 3) 

     

   'Boîte de dialogue 

   MsgBox nom & " " & prenom & ", " & age & " ans" 

End Sub 

Nous allons commencer par ajouter une condition pour vérifier que la valeur de la cellule F5 est numérique avant 

d'exécuter le code. 

 

La fonction IsNumeric sera utilisée dans cette condition : 

Sub variables() 

 

  'SI la valeur entre parenthèses (cellule F5) est numérique (DONC SI CONDITION VRAIE) alors on 

  'exécute les instructions placées après THEN 

   If IsNumeric(Range("F5")) Then 

     

       'Déclaration des variables 

       Dim nom As String, prenom As String, age As Integer, numero_ligne As Integer 

       'Valeurs des variables 

       numero_ligne = Range("F5") + 1 

       nom = Cells(numero_ligne, 1) 

       prenom = Cells(numero_ligne, 2) 

       age = Cells(numero_ligne, 3) 

       'Boîte de dialogue 

       MsgBox nom & " " & prenom & ", " & age & " ans" 

         

    End If 

     

End Sub 

Ajoutons également des instructions pour le cas où la condition n'est pas remplie : 

Sub variables() 

 

    If IsNumeric(Range("F5")) Then 'SI CONDITION VRAIE 

    

       'Déclaration des variables 

       Dim nom As String, prenom As String, age As Integer, numero_ligne As Integer 

       'Valeurs des variables 

       numero_ligne = Range("F5") + 1 

       nom = Cells(numero_ligne, 1) 

       prenom = Cells(numero_ligne, 2) 

       age = Cells(numero_ligne, 3) 

       'Boîte de dialogue 

       MsgBox nom & " " & prenom & ", " & age & " ans" 

         

    Else 'SI CONDITION FAUSSE 

    

       'Boîte de dialogue : avertissement 

       MsgBox "L'entrée " & Range("F5") & " n'est pas valide !" 

       'Suppression du contenu de la cellule F5 

       Range("F5").ClearContents 

     

    End If 

     

End Sub 

Les valeurs non numériques ne sont désormais plus un problème. 

Notre tableau contient 16 lignes de données, nous allons donc vérifier que la variable numero_ligne soit : "plus grande 

ou égale à 2" et "plus petite ou égale à 17". 

Mais avant, voici les opérateurs de comparaison : 

= est égal à 

<> est différent de 

< est plus petit que 

<= est plus petit ou égal à 

> est plus grand que 

>= est plus grand ou égal à 

Ainsi que d'autres opérateurs utiles : 

AND et 
[condition1] AND [condition2] 

Les 2 conditions doivent être vraies 

OR ou 
[condition1] OR [condition2] 

Au moins 1 des 2 conditions doit être vraie 

NOT faux 
NOT [condition1] 

La condition doit être fausse 

Ajoutons maintenant les conditions indiquées un peu plus haut en utilisant AND ainsi que les opérateurs de 

comparaison détaillés ci-dessus : 

Sub variables() 

    If IsNumeric(Range("F5")) Then 'SI NUMERIQUE 

        Dim nom As String, prenom As String, age As Integer, numero_ligne As Integer 

        numero_ligne = Range("F5") + 1 

 

        If numero_ligne >= 2 And numero_ligne <= 17 Then 'SI N° CORRECT 

            nom = Cells(numero_ligne, 1) 

            prenom = Cells(numero_ligne, 2) 

            age = Cells(numero_ligne, 3) 

            MsgBox nom & " " & prenom & ", " & age & " ans" 

        Else 'SI N° INCORRECT 

            MsgBox "L'entrée " & Range("F5") & " n'est pas un numéro valide !" 

            Range("F5").ClearContents 

        End If 

         

    Else 'SI NON NUMERIQUE 

        MsgBox "L'entrée " & Range("F5") & " n'est pas valide !" 

        Range("F5").ClearContents 

    End If 

End Sub 

 

Pour rendre notre macro plus pratique, nous pouvons encore remplacer 17 par une variable contenant le nombre de 

lignes. Cela permettra d'ajouter/retirer des lignes à notre tableau sans avoir à modifier à chaque fois cette limite. 

Pour cela, créons une variable nb_lignes et ajoutons cette fonction : 

 
WorksheetFunction.CountA ne vous dit probablement rien mais il s'agit en fait de la fonction NBVAL que vous 

connaissez probablement déjà (sinon, cliquez ici). 

Nous demandons à cette fonction de comptabiliser le nombre de cellules non vides de la première colonne et nous 

remplaçons ensuite 17 par nb_lignes : 

Sub variables() 

    If IsNumeric(Range("F5")) Then 'SI NUMERIQUE 

        Dim nom As String, prenom As String, age As Integer, numero_ligne As Integer 

        Dim nb_lignes As Integer 

         

        numero_ligne = Range("F5") + 1 

        nb_lignes = WorksheetFunction.CountA(Range("A:A")) 'Fonction NBVAL 

        

        If numero_ligne >= 2 And numero_ligne <= nb_lignes Then 'SI N° CORRECT 

            nom = Cells(numero_ligne, 1) 

            prenom = Cells(numero_ligne, 2) 

            age = Cells(numero_ligne, 3) 

            MsgBox nom & " " & prenom & ", " & age & " ans" 

        Else 'SI N° INCORRECT 

            MsgBox "L'entrée " & Range("F5") & " n'est pas un numéro valide !" 

            Range("F5").ClearContents 

        End If 

 

    Else 'SI NON NUMERIQUE 

        MsgBox "L'entrée " & Range("F5") & " n'est pas valide !" 

http://www.excel-pratique.com/fr/fonctions/nbval.php

        Range("F5").ClearContents 

    End If 

End Sub 

1.2 ElseIf 
ElseIf permet d'ajouter plusieurs conditions à la suite : 

If [CONDITION 1] Then ' => SI condition 1 validée ALORS 

   'Instructions 1 

ElseIf [CONDITION 2] Then ' => SINON, SI condition 2 validée ALORS 

   'Instructions 2 

Else ' => SINON 

   'Instructions 3 

End If 

Si la condition 1 est vraie, les instructions 1 sont exécutées puis nous sortons de l'instruction If (qui débute avec If et 

finit à End If). Si la condition 1 est fausse, nous passons à la condition 2. Si celle-ci est vraie les instructions 2 sont 

exécutées si ce n'est pas le cas les instructions 3 seront alors exécutées. 

Voici un exemple, avec en A1 une note de 1 à 6 (sans virgules pour cet exemple) et en B1 un commentaire en fonction 

de la note : 

Sub commentaires_notes() 

    'Variables 

    Dim note As Integer, commentaire As String 

    note = Range("A1") 

     

    'Commentaire en fonction de la note 

    If note = 6 Then 

        commentaire = "Excellent résultat !" 

    ElseIf note = 5 Then 

        commentaire = "Bon résultat" 

    ElseIf note = 4 Then 

        commentaire = "Résultat satisfaisant" 

    ElseIf note = 3 Then 

        commentaire = "Résultat insatisfaisant" 

    ElseIf note = 2 Then 

        commentaire = "Mauvais résultat" 

    ElseIf note = 1 Then 

        commentaire = "Résultat exécrable" 

    Else 

        commentaire = "Aucun résultat" 

    End If 

     

    'Commentaire en B1 

    Range("B1") = commentaire 

End Sub 

1.3 Select 
Une alternative aux instructions If contenant beaucoup ElseIf existe : Select, cette instruction étant plus adaptée 

dans ce genre de situations. 

Voici la même macro avec Select : 

Sub commentaires_notes() 

    'Variables 

    Dim note As Integer, commentaire As String 

    note = Range("A1") 

     

    'Commentaire en fonction de la note 

    Select Case note    ' <= la valeur à tester (ici, la note) 

    Case Is = 6         ' <= si la valeur = 6 

       commentaire = "Excellent résultat !" 

    Case Is = 5         ' <= si la valeur = 5 

       commentaire = "Bon résultat" 

    Case Is = 4         ' <= si la valeur = 4 

       commentaire = "Résultat satisfaisant" 

    Case Is = 3         ' <= si la valeur = 3 

       commentaire = "Résultat insatisfaisant" 

    Case Is = 2         ' <= si la valeur = 2 

       commentaire = "Mauvais résultat" 

    Case Is = 1         ' <= si la valeur = 1 

       commentaire = "Résultat exécrable" 

    Case Else           ' <= si la valeur n'est égale à aucune des valeurs ci-dessus 

       commentaire = "Aucun résultat" 

    End Select 

     

    'Commentaire en B1 

    Range("B1") = commentaire 

End Sub 

Notez que nous pouvons également utiliser les autres opérateurs de comparaison, par exemple : 

Case Is >= 6         'si la valeur >= 6 

Exemples avec plusieurs valeurs : 

Case Is = 6, 7       'si la valeur = 6 ou 7 

Case Is <> 6, 7      'si la valeur est différente de 6 ou 7 

 

Case 6 To 10         'si la valeur = de 6 à 10 

1.4 Condition en fonction d'un type 
IsNumeric (fonction vue à la page précédente) renvoie TRUE (vrai) si la valeur est numérique et FALSE (faux) si ce 

n'est pas le cas : 

If IsNumeric(Range("A1")) = True Then 'SI LA VALEUR EST NUMERIQUE ... 

Le code suivant est identique au premier (il n'est pas nécessaire d'indiquer = True puisque que l'on cherche 

automatiquement à savoir si la condition est vraie) : 

If IsNumeric(Range("A1")) Then 'SI LA VALEUR EST NUMERIQUE ... 

Dans le cas où nous voulons vérifier si la valeur n'est pas numérique, nous avons également deux possibilités : 

If IsNumeric(Range("A1")) = False Then 'SI LA VALEUR N'EST PAS NUMERIQUE ... 

 

If Not IsNumeric(Range("A1")) Then 'SI LA VALEUR N'EST PAS NUMERIQUE ... 

D'autres fonctions proches de IsNumeric : 

If IsDate(Range("A1")) Then 'SI LA VALEUR EST UNE DATE ... 

 

If IsEmpty(Range("A1")) Then 'SI VIDE ... 

 

If var_objet Is Nothing Then 'SI OBJET NON INITIALISE ... 

1.5 Condition en fonction du type d'une variable 
Pour effectuer des actions en fonction du type d'une variable (Variant), nous aurons besoin de la fonction VarType. 

Après avoir ajouté le signe =, la liste des types apparaît : 

 

If VarType(ma_variable) = vbInteger Then 'SI ma_variable est de type Integer ... 

La valeur des constantes : 

Constante Valeur 

vbEmpty 0 

vbNull 1 

vbInteger 2 

vbLong 3 

vbSingle 4 

vbDouble 5 

vbCurrency 6 

vbDate 7 

vbString 8 

vbObject 9 

vbError 10 

If VarType(ma_variable) = vbInteger Then 'SI ma_variable est de type Integer ... 

'Identique à : 

If VarType(ma_variable) = 2 Then 'SI ma_variable est de type Integer ... 

 

1.6 Condition en fonction de la comparaison de 2 chaînes de caractères 
Jusque-là nous n'avons vu que cela : 

ma_variable = "Exemple 12345" 

 

If ma_variable = "Exemple 12345" Then ' => VRAI 

 

Les 2 chaînes de caractères sont identiques ici, rien d'extraordinaire ... 

Maintenant, si nous voulons vérifier que la variable contienne bien la valeur "12345" sans tenir compte des autres 

caractères, nous utiliserons l'opérateur Like ainsi que * devant et derrière la valeur à rechercher. 

Le caractère * peut remplacer : aucun, un ou plusieurs caractères : 

ma_variable = "Exemple 12345" 

 

If ma_variable Like "*12345*" Then ' => VRAI 

 

Le caractère # peut remplacer un caractère numérique de 0 à 9 : 

ma_variable = "Exemple 12345" 

 

If ma_variable Like "Exemple 12###" Then ' => VRAI 

 

Le caractère ? peut remplacer un caractère quelconque : 

ma_variable = "Exemple 12345" 

 

If ma_variable Like "?xemple?1234?" Then ' => VRAI 

 

Nous pouvons également remplacer un caractère en fonction d'une plage de caractères ou de caractères précis : 

 [abc] remplace un des caractères suivants : a b c 

 [a-g] remplace un des caractères suivants : a b c d e f g 

 [369] remplace un des caractères suivants : 3 6 9 

 [2-5] remplace un des caractères suivants : 2 3 4 5 

 [?*#] remplace un des caractères suivants : ? * # 

ma_variable = "Exemple 12345" 

 

If ma_variable Like "[BIEN]xemple 1234[4-7]" Then ' => VRAI 

 

Pour remplacer un caractère non compris dans les valeurs entre crochets, un ! doit être ajouté après [ : 

ma_variable = "Exemple 12345" 

If ma_variable Like "[!FAUX]xemple 1234[!6-9]" Then ' => VRAI 

 

Remarque : un caractère en majuscule n'est pas égal à ce même caractère en minuscule. Pour ne pas faire de 

distinctions entre majuscules-minuscules, placez Option Compare Text en début de module. 

2 Les boucles 

2.1 While 
Les boucles permettent de répéter des instructions un certain nombre de fois, ce qui peut être un gain de temps 

considérable. 

Le code suivant numérote les cellules de la colonne A (de la ligne 1 à 12) : 

 

Sub boucle_while() 

 

    Cells(1, 1) = 1 

    Cells(2, 1) = 2 

    Cells(3, 1) = 3 

    Cells(4, 1) = 4 

    Cells(5, 1) = 5 

    Cells(6, 1) = 6 

    Cells(7, 1) = 7 

    Cells(8, 1) = 8 

    Cells(9, 1) = 9 

    Cells(10, 1) = 10 

    Cells(11, 1) = 11 

    Cells(12, 1) = 12 

 

End Sub 

Ce code est très répétitif ... 

Maintenant, imaginez qu'il faille numéroter plusieurs centaines de lignes ... Vous comprenez donc l'intérêt de créer 

des boucles. 

Voici la boucle While : 

Sub boucle_while() 

 

    While [condition] 

        'Instructions 

    Wend 

 

End Sub 

Tant que la condition est vraie, les instructions sont exécutées en boucle (attention à ne pas créer une boucle infinie). 

Voici la macro répétitive ci-dessus avec la boucle While : 

Sub boucle_while() 

 

    Dim numero As Integer 

    numero = 1 'Numéro de départ (correspond ici au n° de ligne et au n° de numérotation) 

    While numero <= 12 'TANT QUE la variable numero est <= 12, la boucle est répétée 

       Cells(numero, 1) = numero 'Numérotation 

       numero = numero + 1 'Le numéro est augmenté de 1 à chaque boucle 

    Wend 

         

End Sub 

Avec cette boucle, si nous voulons numéroter 500 lignes, il suffit alors de remplacer 12 par 500 ... 

Do Loop 

Cette boucle fonctionne de la même manière que While Wend (tant que la condition est vraie, la boucle est exécutée) 

: 

Sub boucle_do_while() 

 

    Do While [condition] 

        'Instructions 

    Loop 

 

End Sub 

La condition peut également être placée en fin de boucle Do Loop, ce qui implique que les instructions sont exécutées 

au moins une fois : 

Sub boucle_do_while() 

 

    Do 

        'Instructions 

    Loop While [condition] 

 

End Sub 

Plutôt que de répéter la boucle tant que la condition est vraie, il est possible de quitter la boucle lorsque la condition 

est vraie en remplaçant While par Until : 

Sub boucle_do_while() 

 

    Do Until [condition] 

        'Instructions 

    Loop 

 

End Sub 

2.2 For 

Sub boucle_for() 

 

    For i = 1 To 5 

        'Instructions 

    Next 

 

End Sub 

La boucle For est répétée ici 5 fois. 

A chaque répétition de la boucle, la variable i est automatiquement augmentée de 1 : 

Sub boucle_for() 

 

    For i = 1 To 5 

        MsgBox i 

    Next 

 

End Sub 

2.3 Quitter une boucle prématurément 
Il est possible de quitter une boucle For prématurément grâce à l'instruction suivante : 

Exit For 'Quitter une boucle For 

 

En voici un exemple : 

Sub boucle_for() 

    Dim max_boucles As Integer 

    max_boucles = Range("A1") 'En A1 : une limite de répétitions de la boucle est définie 

 

    For i = 1 To 7 'Boucles prévues : 7 

       If i > max_boucles Then 'Si A1 est vide ou contient un nombre < 7, diminution du nb de boucles 

           Exit For 'Si condition vraie, on quitte la boucle For 

       End If 

     

       MsgBox i 

    Next 

 

End Sub 

Les autres instructions Exit : 

Exit Do 'Quitter une boucle Do Loop 

 

Exit Sub 'Quitter une procédure 

 

Exit Function 'Quitter une fonction 

3 Procédures et fonctions 

3.1 Public - Private 
Pour le moment, toutes les procédures créées sont de typePublic, elles sont accessibles depuis tous les modules. 

Sub exemple() 

'Identique à : 

Public Sub exemple() 

Pour rendre une procédure inaccessible hors du module, ajoutez Private : 

Private Sub exemple() 

3.2 Lancer une procédure depuis une procédure 
Pour exécuter une procédure depuis une autre procédure, entrez simplement son nom. 

Un exemple simple : 

 

Private Sub avertissement() 

    MsgBox "Attention !!!" 

End Sub 

 

 

Sub macro_test() 

    If Range("A1") = "" Then 

        avertissement ' <= exécute la procédure "avertissement" 

    End If 

    'etc ... 

End Sub 

Ici, lorsque "macro_test" est exécutée et que A1 vaut "", la procédure "avertissement" est exécutée. 

3.3 Les arguments 
Les arguments permettent d'utiliser des valeurs d'une procédure dans une sous-procédure (car rappelez-vous que 

par défaut les variables ne sont pas accessibles depuis les autres procédures). 

Private Sub avertissement(texte As String) 

    MsgBox "Attention : " & texte & " !" 

End Sub 

 

 

Sub macro_test() 

    If Range("A1") = "" Then 

        avertissement "cellule vide" 

    ElseIf Not IsNumeric(Range("A1")) Then 

        avertissement "valeur non numérique" 

    End If 

End Sub 

 
Un argument a été ajouté à la procédure "avertissement", il s'agit de la variable "texte" de type "String" : 

Private Sub avertissement(texte As String) 

Cette procédure nécessite un argument, il faudra donc placer une valeur après "avertissement" pour l'exécuter : 

avertissement "cellule vide" 

En cas d'arguments multiples, ceux-ci doivent être séparés par des virgules. 

3.4 Les arguments optionnels 
Par défaut, si une procédure requiert des arguments, ceux-ci sont obligatoires pour exécuter la procédure. 

Des arguments optionnels peuvent être ajoutés après les arguments obligatoires avec Optional, par exemple : 

Private Sub boite_de_dialogue(nom As String, Optional prenom, Optional age) 

 

Cette procédure peut alors être lancée avec ou sans arguments optionnels, comme ceci : 

'Exemple 1 : on affiche le nom : 

 

boite_de_dialogue nom1 

     

'Exemple 2 : on affiche le nom et le prénom : 

 

boite_de_dialogue nom1, prenom1 

     

'Exemple 3 : on affiche le nom et l'âge : 

 

boite_de_dialogue nom1, , age1 

     

'Exemple 4 : on affiche le nom, le prénom et l'âge : 

 

boite_de_dialogue nom1, prenom1, age1 

Les arguments doivent être indiqués dans l'ordre. 

Pour vérifier si un argument optionnel est présent ou non, nous utiliserons la fonction IsMissing. Cette fonction 

n'étant compatible qu'avec certains types de variables (dont Variant), le type des arguments optionnels n'a pas été 

déclaré (type non déclaré = Variant). 

Voici un exemple avec les 2 portions de code ci-dessus : 

Sub macro_test() 

 

    Dim nom1 As String, prenom1 As String, age1 As Integer 

     

    nom1 = Range("A1") 

    prenom1 = Range("B1") 

    age1 = Range("C1") 

 

    'Exemple 1 : on affiche le nom : 

    boite_de_dialogue nom1 

     

    'Exemple 2 : on affiche le nom et le prénom : 

    boite_de_dialogue nom1, prenom1 

     

    'Exemple 3 : on affiche le nom et l'âge : 

    boite_de_dialogue nom1, , age1 

     

    'Exemple 4 : on affiche le nom, le prénom et l'âge : 

    boite_de_dialogue nom1, prenom1, age1 

 

End Sub 

 

Private Sub boite_de_dialogue(nom As String, Optional prenom, Optional age) 

     

    If IsMissing(age) Then 'Si la variable age est absente ... 

        

        If IsMissing(prenom) Then 'Si la variable prenom est absente, on n'affiche que le nom 

           MsgBox nom 

        Else 'Sinon, on affiche le nom et le prénom 

           MsgBox nom & " " & prenom 

        End If 

    Else 'Si la variable age est présente ... 

 

        If IsMissing(prenom) Then 'Si la variable prenom est absente, on affiche le nom et l'âge 

           MsgBox nom & ", " & age & " ans" 

        Else 'Sinon on affiche le nom, le prénom et l'âge 

           MsgBox nom & " " & prenom & ", " & age & " ans" 

        End If 

     

    End If 

        

End Sub 

Aperçu (exemple 1) : 

 

3.5 ByRef - ByVal 
Par défaut, les arguments sont de type ByRef ce qui signifie que, si une variable est passée en argument, c'est sa 

référence qui est transmise. Autrement dit, si la variable est modifiée dans la sous-procédure, elle le sera également 

dans la procédure d'appel. 

Par exemple : 

Sub macro_test() 

    Dim nombre As Integer 

    nombre = 30 

     

    calcul_carre nombre 

     

    MsgBox nombre 

End Sub 

 

Private Sub calcul_carre(ByRef valeur As Integer) 'Il n'est pas nécessaire de préciser ByRef (puisque 

par défaut) 

    valeur = valeur * valeur 

End Sub 

Pour mieux comprendre, voici ce qui se passe lorsque la macro est lancée : 

nombre = 30 

'La valeur initiale de la variable "nombre" est 30 

 

calcul_carre nombre 

'La sous procédure est lancée avec la variable "nombre" en argument 

 

Private Sub calcul_carre(ByRef valeur As Integer) 

'La variable "valeur" est en quelque sorte un raccourci vers la variable "nombre", par conséquent, si la 

variable "valeur" est modifiée, cela modifie la variable "nombre" (il n'est pas nécessaire de les nommer 

de façon identique) 

valeur = valeur * valeur 

'La valeur de la variable "valeur" est modifiée (donc la variable "nombre" est modifiée) 

 

End Sub 

'Fin de la sous-procédure 

 

MsgBox nombre 

'La variable "nombre" a été modifiée, 900 est alors affiché dans la boîte de dialogue 

 

La seconde possibilité consiste à utiliser ByVal. 

Contrairement à ByRef qui transmet la référence (raccourci), ByVal transmet la valeur, ce qui signifie que la variable 

passée en argument ne subit aucune modification. 

Voici ce qui se passe avec le code précédent et ByVal : 

nombre = 30 

'La valeur initiale de la variable "nombre" est 30 

 

calcul_carre nombre 

'La sous procédure est lancée avec la variable "nombre" en argument 

 

Private Sub calcul_carre(ByVal valeur As Integer) 

'La variable "valeur" copie la valeur de la variable "nombre" (les 2 variables ne sont pas liées) 

 

valeur = valeur * valeur 

'La valeur de la variable "valeur" est modifiée 

 

End Sub 

'Fin de la sous-procédure (dans cet exemple, la sous-procédure n'aura servi à rien) 

 

MsgBox nombre 

'La variable "nombre" n'a pas été modifiée, 30 est donc affiché dans la boîte de dialogue 

 

Ce qu'il faut retenir : utiliser ByVal lorsque la variable ne doit pas être modifiée ... 

3.6 Les fonctions 
La principale différence entre une procédure Sub et Function est la valeur retournée par la fonction. 

En voici un exemple simple : 

Function carre(nombre) 

    carre = nombre ^ 2 'La fonction "carre" renvoie la valeur de "carre" 

End Function 

 

Sub macro_test() 

    Dim resultat As Double 

    resultat = carre(9.876) 'La variable resultat reçoit la valeur retournée par la fonction 

    MsgBox resultat 'Affiche le résultat (ici, le carré de 9.876) 

End Sub 

Les fonctions peuvent également être utilisées sur la feuille comme n'importe quelle fonction Excel. 

Par exemple, pour obtenir le carré de la valeur de A1 : 

 



 
 

  
BAT 3 / 2016-2017 
H. RENARD-FERRY 

            BAT3 - 2016-2017  C2I 

 

16 
16 

4 Exercices 

4.1 Exercice n°1 : miroirs 
Le miroir d’une chaîne de caractères s est la chaîne formée des mêmes caractères que s, mais l’ordre inverse. Par 

exemple, « nipal » est le miroir de « lapin ». 

Le miroir d’un entier n est le nombre dont l’écriture décimale est formée des mêmes chiffres que n mais dans l’ordre 

inverse. Par exemple, 3156 est le miroir de 6513. 

Question 1:  Écrire une définition récursive de la fonction miroirChaine qui, étant donnée une chaîne de 

caractères s, renvoie le miroir de s. 

Question 2:  Écrire une définition itérative de la fonction miroirChaine. 

Question 3:  Écrire une définition itérative de la fonction miroirNombre qui, étant donné un entier naturel 

n, renvoie le miroir de n. 

 

Mémento :  
Function Left(String As String, Length As Long) As String 

‘Renvoie : la sous-chaine préfixe de la chaine donnée String de longueur Length 

‘(renvoie la chaine String lorsque Length est plus grand que la longueur de String) 

 

Function Mid(String As String, Start As Long, Optional Length As Long) – As String 

‘Renvoie : la sous-chaine, de longueur Length, extradite de la chaine donnée String, 

à partir de l’indice Start (lorsque Length est omis ou lorsque le nombre de caractères 

du texte (à partir et y compris le caractère qui occupe la position de Start), est 

inférieur à la valeur de Length, elle renvoie tous les caractères à partir de Start) 

4.2 Exercice n°2 : carrés concentriques 

L’objectif de cet exercice est de dessiner des carrés concentriques. Pour implémenter cet exercice sous Excel, il faut 

commencer par mettre les cellules de la feuille de calcul, sous forme de petits carrés, comme les figures qui suivent ; 

pour cela, dans le menu Format du tableur, choisir de modifier la hauteur de ligne et la largeur de colonne (on pourra 

prendre par exemple une hauteur de ligne 8 et une largeur de colonne de 1). 

Pour définir les procédures demandées, il est très utile d’utiliser les procédures colorierRectangle et 

supprimerCouleurFeuille dont voici les spécifications : 

Sub colorierRectangle (ByVal i1 As Long, ByVal j1 As Long,  

                       ByVal i2 As Long, ByVal j2 As Long, ByVal c As Long) 

‘Modification : colorie, avec la couleur c, le rectangle ayant pour coins opposés, 

‘les cellules (i1, j1) et (i2, j2) de la feuille active. 

 

Sub supprimerCouleurFeuille() 

‘Modification : supprime les couleurs de la feuille active. 

Question 1: Écrire une définition de la procédure colorierContour qui, étant donnés quatre entiers x0, 

y0, d, et c, colorie en couleur c le contour d’un carré de centre la cellule (x0, y0) et de côté 2d+1. 

 

Tester cette procédure : 

 Avec c = RGB(0,0,0), c’est-à-dire en noir, d=3 et la cellule située en ligne 5 et en colonne 5 comme 

centre, comme le dessin ci-dessous ; 

 

 Avec c = RGB(0,0,0) et en demandant à l’utilisateur d’autres valeurs pour les coordonnées du centre 

et pour d. 

Lors de l’exécution de la macro, tester alors le cas où les données ne sont pas valides et le cas où le contour 

du carré ne tient pas dans la feuille de calcul. 

 

Question 2: Écrire une procédure colorierCarresConcentriques, à trois paramètres x0, y0 et n, qui 

dessine n carrés concentriques noirs de centre la cellule (x0, y0), le carré central étant de couleur noir. 

Tester cette procédure de façon similaire au test demandé dans la question 1. 

 

  

4.3 Exercice n°3 : « little Professor » 

Quand vous étiez enfants, vous aviez peut-être un jouet qui vous entrainait à faire du calcul mental en vous proposant 

des opérations au hasard. C’est ce genre de jeu que je vous propose d’implémenter dans cet exercice. 

Le matériel est composé de : 

 Une boite à calculs, représentée dans la feuille de calcul par une plage de cellules nommée « calculette », 

formée des cellules nommées « opérande1 », « opérande2 », « opérande », « égal » et « résultat », le fond de 

cette plage est teinté en jaune ; 

 Une boite des scores, représentée dans la feuille de calcul par une cellule nommée « note » et une cellule 

nommée « bareme ». 

 

Le jeu se déroule de la façon suivante : 

 Le joueur choisit un niveau de difficulté ; 

 Des calculs lui sont proposés. Pour chaque calcul : 

o L’opération (addition, multiplication, soustraction ou division entière) est tirée au hasard ; 

o Les opérandes sont tirés au hasard, en tenant compte du niveau choisi ; 

o Le barème, qui comptabilise le nombre de calculs déjà proposés, est mis à jour ; 

o La note, qui augmente avec le nombre de bonnes réponses déjà données, est mis à jour ; 

o Le taux de bonnes réponses déjà données est signalé par la couleur de la cellule « note ». 

 Lorsque la série de calculs est terminée, une appréciation est donnée au joueur, tenant compte de sa note et 

du niveau de difficulté. 

 

La figure suivante donne un aperçu du jeu dans Excel. 

 

4.3.1 Fonctions et procédures utiles 

Question 1: Écrire une définition de la fonction opAlea qui renvoie un signe d’opération tiré au hasard parmi 

« + », « * », « - » et « / ». Pour cela, on utilisera la fonction aleatoire qui renverra un entier compris 

entre 1 et 4 (chaque entier pourra ensuite être associé à un des quatre opérateurs), et dont la spécification 

est : 

Function opAlea As String 

‘Renvoie un signe d’opération tiré au hasard parmi « + », « * », « - » et 

« / » 

On utilisera la fonction aleatoire, définie par : 

Function aleatoire (ByVal a As Integer, ByVal b As Integer) As Integer 

‘Renvoie un entier tire au hazard entre a et b compris 

  Aleatoire = Int ((b – a + 1) * Rnd + a) 

End Function 

 

Question 2: Écrire une définition de la fonction resultat qui, étant donnés un signe d’opération parmi 

« + », « * », « - » et « / », et deux opérandes (des entiers) renvoie le résultat de l’opération appliquée aux 

opérandes. 

 

4.3.2 Initialisations 

Le module contient les déclarations suivantes : 

Public Const niveauMax = 9 

Public Const max = 5 

Public Const seuil1 = 0.9 

Public Const seuil2 = 0.5 

Ainsi que la fonction orange : 

Function orange As Long 

Renvoie la couleur orange  

 orange = RGB(255, 120, 0) 

End Function 

 

Question 3:  Le niveau de difficulté varie de 0 à 9 : au niveau 0 les opérandes sont tirés aléatoirement entre 1 

et 10, au niveau 1, ils sont tirés aléatoirement entre 1 et 20 … au niveau 9, ils sont tirés aléatoirement entre 

1 et 100. 

Écrire des fonctions : 

 choixNiveau qui demande au joueur le niveau qu’il souhaite, et renvoie cette valeur ; 

 borneMax qui, étant donné un niveau n (entier), renvoie l’extrémité haute de l’intervalle dans 

lequel les opérandes seront tirés aléatoirement. 

Question 4: Écrire une définition de la procédure initialiser qui initialise le générateur aléatoire, efface toutes 

les valeurs contenues dans la plage « calculette » et met la valeur à 0 dans les cellules « note » et 

« bareme ». 

 

4.3.3 Le jeu 

Question 5: Écrire une définition de la procédure prepareCalcul qui, étant donné un entier b, tire un signe 

d’opération et des opérandes (comprises entre 1 et b) au hasard, et remplit les cellules de la boite à calculs 

en utilisant ces données. Elle doit aussi calculer le résultat de l’opération et le mettre dans la cellule 

« resultat », sans qu’il soit visible. Cette procédure aura pour paramètre l’extrémité haute de l’intervalle 

dans lequel les nombres seront tirés aléatoirement. 

Attention à la soustraction (le résultat ne doit pas être négatif) et à la division (le quotient doit être 

entier). 

Question 6: Écrire une définition de la procédure repondreEtAjusterScores qui :  

 Saisit la réponse du joueur dans une boite de dialogue, en autorisant au plus deux essais ; 

 Augmente la note de 1 si la réponse est bonne dès le premier essai, de 0,5 si la réponse est bonne 

au deuxième essai, ne la modifie pas si les deux premières réponses sont incorrectes ; 

 Signale (dans une boite de message) si la réponse est correcte ou non ; 

 Rend visible (dans la cellule « resultat » le résultat obtenu ; 

 Met à jour le contenu des cellules « note » et « bareme » 

Question 7:  Le taux de réussite est égal à la note du joueur divisée par le nombre de calculs déjà proposés. 

On désire attirer l’attention du joueur sur son taux de réussite, en coloriant la cellule « note » en 

 Rouge si le taux est médiocre (inférieur à seuil1) ; 

 Orange si le taux est moyen (compris entre seuil1 compris et seuil2 exclu) ; 

 Vert si le taux est bon (supérieur ou égal à seuil2). 

Question 8:  Écrire une définition de la procédure jouer qui, étant donné un entier b propose, nbMax calculs 

aléatoires avec des opérandes compris entre 1 et b, et fait les mises à jour induites. 

Question 9: Écrire une définition de la procédure apprecier qui, étant donné un entier niveau donne une 

appréciation dans une boite de message, en tenant compte de la note obtenue et du niveau niveau. 

Question 10: Écrire une définition de la procédure lancerNouveauJeu qui 

 Fait les initialisations nécessaires ; 

 Fait jouer l’utilisateur (en lui proposant nbMax calculs aléatoires) ; 

 Lui donne une appréciation. 

 


